

ARMY RESEARCH LABORATORY



AESOP Field Prediction

by Brian B. Luu and Calvin D. Le

ARL-TR-835

November 1995

19951211 043



DTIC QUALITY INSPECTED 1

Approved for public release; distribution unlimited.

The findings in this report are not to be construed as an official Department of the Army position unless so designated by other authorized documents.

Citation of manufacturer's or trade names does not constitute an official endorsement or approval of the use thereof.

Destroy this report when it is no longer needed. Do not return it to the originator.

REPORT DOCUMENTATION PAGE			Form Approved OMB No. 0704-0188	
Public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302, and to the Office of Management and Budget, Paperwork Reduction Project (0704-0188), Washington, DC 20503.				
1. AGENCY USE ONLY (Leave blank)		2. REPORT DATE November 1995		3. REPORT TYPE AND DATES COVERED Final, from Oct. 1993–Sept. 1994
4. TITLE AND SUBTITLE AESOP Field Prediction			5. FUNDING NUMBERS PE: 62120	
6. AUTHOR(S) Brian B. Luu and Calvin D. Le				
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) U.S. Army Research Laboratory Attn: AMSRL-WT-ND 2800 Powder Mill Road Adelphi, MD 20783-1197			8. PERFORMING ORGANIZATION REPORT NUMBER ARL-TR-835	
9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES) U.S. Army Research Laboratory 2800 Powder Mill Road Adelphi, MD 20783-1197			10. SPONSORING/MONITORING AGENCY REPORT NUMBER	
11. SUPPLEMENTARY NOTES AMS code: 622120.H250011 ARL PR: 5FE7E2				
12a. DISTRIBUTION/AVAILABILITY STATEMENT Approved for public release; distribution unlimited.			12b. DISTRIBUTION CODE	
13. ABSTRACT (Maximum 200 words) A model to predict the electromagnetic fields emitted by AESOP (Army Electromagnetic Pulse Simulator Operation) is presented. Transmission-line theory is used to develop the antenna currents, and electric dipole arrays are employed to calculate the radiated field. The dipole arrays, representing AESOP, are placed over a perfectly conducting ground. A computer code was developed to estimate the emitted field signature of AESOP.				
14. SUBJECT TERMS EMP simulators, dipole models, antenna modeling			15. NUMBER OF PAGES 51	
			16. PRICE CODE	
17. SECURITY CLASSIFICATION OF REPORT Unclassified	18. SECURITY CLASSIFICATION OF THIS PAGE Unclassified	19. SECURITY CLASSIFICATION OF ABSTRACT Unclassified	20. LIMITATION OF ABSTRACT UL	

Contents

1. Introduction	5
2. AESOP Simulator	6
3. Theory	7
3.1 Current Distribution	8
3.2 Electric Field and Magnetic Field of a Dipole	11
4. Implementation	14
5. Results and Comparison with Empirical Data	16
6. Conclusion	17
Distribution	53

Appendices

A.—Program Listing	19
B.—Field Plot Comparison	29

Figures

1. AESOP at WRF	5
2. General layout of AESOP system	6
3. AESOP characteristics	7
4. Circuit models	8
5. Current comparisons	11
6. Dipole model	12
7. Illustration of dipole model for AESOP simulator	15
8. Coordinates for slanted dipoles	15

Accession For	
NTIS GRA&I	<input checked="" type="checkbox"/>
DTIC TAB	<input type="checkbox"/>
Unannounced	<input type="checkbox"/>
Justification	
By _____	
Distribution/ _____	
Availability Codes	
Dist	Avail and/or Special
A-1	

1. Introduction

Under the sponsorship of the Defense Nuclear Agency (DNA), the Physics International Company developed a Transportable Electromagnetic Pulse Simulator (TEMPS) for Harry Diamond Laboratories (HDL)* in 1972. In 1974, HDL procured a copy of TEMPS, called AESOP (Army electromagnetic pulse (EMP) Simulator Operation), which was erected at the Woodbridge Research Facility (WRF) (see fig. 1). In 1994, AESOP was transferred to the Test and Evaluation Command (TECOM) at White Sands Missile Range. The main function of AESOP is to simulate the electromagnetic environment produced by an exoatmospheric nuclear burst. By simulating the threat-level EMP environment, it allows for testing and verifying the EMP susceptibility or hardness of a system. Since AESOP is a virtual copy of TEMPS (except for transportability and some minor modifications and enhancements), the specifications of AESOP are documented in the TEMPS's technical manuals.

Because of the complexity of EMP simulation and the limited computing capability in the 70s, no computing codes were developed to calculate or approximate the field generated by AESOP. A field mapping report¹ of TEMPS was developed to provide the EMP fields at key locations near the ground. The present report provides results from a code that calculates the electromagnetic (EM) field generated by AESOP (the code is given in app A).



Figure 1. AESOP at WRF.

*Now part of the Army Research Laboratory.

¹Eugene L. Patrick and Spencer L. Soohoo, *Transportable Electromagnetic Pulse Simulator (TEMPS) Field Mapping Report*, Harry Diamond Laboratories, HDL-TR-1743 (Reprinted May 1984).

2. AESOP Simulator

Most of the information in this section is derived from the Physics International manual for TEMPS.²

Located at WRF in a 120,000-m² test area, AESOP is the largest EMP free-field simulator in the world. It consists of three basic subsystems: a support structure, pulse generator, and antenna.

- The support structure, made of fiberglass columns and laminated wood beams, is used to support and suspend the pulse generator and antenna at a height up to 20 m above the ground.
- The pulse generator mainly comprises the pulser assembly with two Marx generators, peaking capacitors, output switch, and control subsystem.
- The antenna is composed of a cylindrical and conical system of 36 wires extending from each end of the pulser assembly with associated forming hoops, end assemblies, tensioning elements, and end grounding terminations. The diameter of the cylindrical section is 9.144 m.

The general layout of the AESOP system is shown in figure 2.² The overall length of AESOP is 300 m from end dome to end dome. The pulser and antenna are normally positioned at 20 m above the ground level but can also be at lower elevations.





Figure 2. General layout of AESOP system.

²Physics International Company, *Transportable Electromagnetic Pulse Simulator System (TEMPS) Operation & Maintenance Manual*, PIMM-372 (April 1973).

AESOP produces a pulsed EM field by transmitting a high-energy electrical pulse generated from the pulse generator to the ground through the antenna. It can provide a radiated free-field range of 20 to 50 kV/m from the pulser on the center line. The pulse shape is a double exponential with less than 10-percent undershoot (fig. 3). For more information, see the TEMPS manual.²

Figure 3. AESOP characteristics:
(a) pulse shape and
(b) span on EMP coverage.



3. Theory

In the model presented here, the AESOP antenna is modeled as a series of dipole antenna elements above a perfectly conducting ground. There are two steps involved in the calculation of the radiated electromagnetic field from AESOP. A closed-form analytical time-domain current profile on each dipole element is derived by the transmission-line method. (In order to simplify the problem, we account only for reflection at the load end of AESOP.) The time-domain radiated electromagnetic field from each dipole is then computed by a convolution technique.

²Physics International Company, *Transportable Electromagnetic Pulse Simulator System (TEMPS) Operation & Maintenance Manual*, PIMM-372 (April 1973).

3.1 Current Distribution


AESOP is represented by a transmission line over a perfectly conducting ground. There are three factors in this transmission-line model that need to be considered.²

First, the 35-stage Marx generator can be modeled in a lumped equivalent circuit, as shown in figure 4(a). The switches S_1 and S_2 are the voltage charging and discharging switches, respectively. Z_a is the characteristic impedance of the antenna, which includes the biconical and conical sections. Z_l is the termination load.

Second, the biconical section with cones at angle $\theta = 40.5^\circ$ has a characteristic impedance, Z_b ,

$$Z_b = \frac{Z_0}{\pi} \ln \left[\cot \left(\frac{\theta}{2} \right) \right] \approx 120 \, \Omega, \quad (1)$$

Figure 4. Circuit models: (a) AESOP, (b) transmission-line model, and (c) AESOP model at $t = t_2$.



²Physics International Company, Transportable Electromagnetic Pulse Simulator System (TEMPS) Operation & Maintenance Manual, PIMM-372 (April 1973).

where

$$Z_0 = \sqrt{\frac{\mu_0}{\epsilon_0}} \approx 377 \Omega .$$

Third, the cylindrical system of wires that makes up the AESOP antenna structure can be treated as low-loss transmission line of a wire over the ground plane. Its characteristic impedance, Z_c , is approximated by R , L , and C (R is series resistance per unit length, L is series inductance per unit length, and C is shunt capacitance per unit length):

$$Z_c = \left(1 + \frac{R}{2sL}\right) \sqrt{\frac{L}{C}} \approx 130 \left(1 + \frac{R}{2sL}\right) , \quad (2)$$

where

$$L = \frac{\mu_0}{2\pi} \ln\left(\frac{2h}{a}\right) ,$$

$$C = \frac{2\pi\epsilon_0}{\ln\left(\frac{2h}{a}\right)} ,$$

$$s = j\omega ,$$

$$\omega = \text{angular frequency},$$

$$h = \text{height of a wire over the ground plane, and}$$

$$a = \text{radius of a wire}.$$

The conical section is designed to maintain the same characteristic impedance as the cylindrical section. In other words, the ratio of $2h/a$ remains constant for both conical and cylindrical sections.

For the low-resistive-loss biconical section, the characteristic impedance, Z_b , is very compatible with the characteristic impedance, Z_c , of the cylindrical section. Since the cylindrical section is much longer than the biconical section, the assumption is that the characteristic impedance Z_b equals the characteristic impedance Z_c . The schematic representation of a 1-cm-unit-length section, δl , of Z_c is shown in figure 4(b).

At time $t = t_2$, we obtain the Laplace transform for the AESOP model (fig. 4(c)):

$$V_0 = V_3(t_2) \left(\frac{Z_l \sqrt{\frac{L}{C}}}{L_2 \left(Z_l + \sqrt{\frac{L}{C}} \right)} \right) \left(\frac{\prod_{i=1}^3 (s - y_i)}{\prod_{i=1}^5 (s - s_i)} \right) , \quad (3)$$

where

V_1, V_2, V_3 are initial voltage sources at $t = t_2^-$,

Z_{eq} is the equivalent impedance of R_2, L_2, Z_b, Z_c , and Z_l ,

y_1, y_2 are roots of

$$y^2 + \left[\frac{V_2(t_2)}{V_3(t_2)C_2} + \frac{R_1}{L_1} \right] y + \left[\frac{V_1(t_2)}{V_3(t_2)L_1C_2} + \frac{1}{L_1C_1} \right] = 0 ,$$

$$y_3 = \frac{R}{2L} ,$$

s_1, s_2, s_3, s_4, s_5 are roots of

$$s^5 + A_1s^4 + A_2s^3 + A_3s^2 + A_4s + A_5 = 0 , \text{ and}$$

A_1, A_2, A_3, A_4, A_5 are functions of $R, R_1, R_2, L, L_1, L_2, C, C_1, C_2$.

The reflecting voltage coefficient at the load end is

$$\Gamma = \frac{Z_l - Z_c}{Z_l + Z_c} = |\Gamma| e^{j\phi} . \quad (4)$$

Voltage across the capacitor, V_n , at any given n^{th} section of the antenna is calculated as

$$V_n = V_0 [e^{-n\psi} + \Gamma e^{-(2N-n)\psi}] , \quad (5)$$

where

$$\begin{aligned} \psi &= \alpha + j\beta , \\ \alpha &= \frac{R}{2\sqrt{\frac{L}{C}}} \delta l , \\ \beta &= \omega \sqrt{\frac{L}{C}} \delta l . \end{aligned}$$

The corresponding current on the n^{th} section of the antenna is

$$I_n = (V_n - V_{n+1}) \frac{1}{L} \left(\frac{1}{s + \frac{R}{L}} \right) . \quad (6)$$

The time-domain current, $I_n(t)$, can be calculated directly from the above equation by the Laplace transform technique:

$$I_n(t) = \sum_{k=1}^{14} [B_{nk} \exp(D_{nk}t)] u(t - t_n) , \quad (7)$$

where

B_{nk} is the time constant coefficient number for the k^{th} variable and the n^{th} section,

D_{nk} is the time constant exponential coefficient number for the k^{th} variable and the n^{th} section, and


t_n is the delay time constant for the n^{th} section.

The calculated current distributions at various locations on the AESOP antenna are shown in figure 5(a). Figure 5(b) compares the results of this method to those of the finite time-difference method. Good agreement was obtained at the early time, and reasonable agreement was obtained at the later time.

3.2 Electric Field and Magnetic Field of a Dipole

As described in section 2, an electromagnetic field is emitted from the AESOP antenna, which is a transmission-line antenna. It can be modeled as a series of linear dipoles. Each dipole itself is an elementary antenna that radiates fields. In this section, the electric field and magnetic field (radiated free field) equations of a dipole model are expanded from section 2.3 of

Figure 5. Current comparisons: (a) on different locations away from center of AESOP and (b) on bicone and at 1.3 ft away from center.



Collin³ and used as the basis for computing the fields of the AESOP simulator.

We begin with the hypothesis of a short current filament (dl), which can be represented as a dipole, located at the origin and oriented along the z -axis as shown in figure 6(a). The magnetic and electric fields generated by the dipole at an observation point are given by


$$\vec{E} = \frac{Z_0 dl}{2\pi} \left(\frac{1}{r^2} + \frac{1}{jk_0 r^3} \right) I e^{-jk_0 r} \cos \theta \hat{r} + \frac{Z_0 dl}{4\pi} \left(\frac{jk_0}{r} + \frac{1}{r^2} + \frac{1}{jk_0 r^3} \right) I e^{-jk_0 r} \sin \theta \hat{\theta} , \quad (8)$$

$$\vec{H} = \frac{dl}{4\pi} \left(\frac{jk_0}{r} + \frac{1}{r^2} \right) I e^{-jk_0 r} \sin \theta \hat{\phi} , \quad (9)$$

where

$$\begin{aligned} Z_0 &= (\mu_0 / \epsilon_0)^{1/2} \equiv \text{intrinsic impedance of free space,} \\ k_0 &= \omega (\mu_0 \epsilon_0)^{1/2} \equiv \text{free-space wave number, and} \\ c &= (\mu_0 \epsilon_0)^{-1/2} \equiv \text{speed of light in free space.} \end{aligned}$$

Figure 6. Dipole model: (a) original, (b) general coordinates, and (c) rotate coordinates.



³Robert E. Collin, *Antennas and Radiowave Propagation*, McGraw-Hill Book Company, 1985.

In the more general case shown in figure 6(b), $r(x,y,z)$ and $r'(x',y',z')$ are the coordinates of the observation point and of the dipole, respectively. Hence, $R(X,Y,Z)$ is the relative coordinate of the observation point in the Cartesian coordinate with the origin at the dipole, where $\vec{R} = \vec{r} - \vec{r}'$. The magnitude of R can be defined as

$$R = \sqrt{(x-x')^2 + (y-y')^2 + (z-z')^2} .$$

In the general case, then, equations (8) and (9) become

$$\begin{aligned} \bar{E}(\omega) = & \frac{Z_0 dl}{2\pi} \left(\frac{1}{R^2} + \frac{c}{j\omega R^3} \right) I(\omega) e^{-j\omega R/c} \cos \theta \hat{R} \\ & + \frac{Z_0 dl}{4\pi} \left(\frac{j\omega}{cR} + \frac{1}{R^2} + \frac{c}{j\omega R^3} \right) I(\omega) e^{-j\omega R/c} \sin \theta \hat{\Theta} , \end{aligned} \quad (10)$$

$$\bar{H}(\omega) = \frac{dl}{4\pi} \left(\frac{j\omega}{cR} + \frac{1}{R^2} \right) I(\omega) e^{-j\omega R/c} \sin \theta \hat{\Phi} . \quad (11)$$

By transferring equations (10) and (11) into the time domain and changing to Cartesian coordinates, we transform equations (10) and (11) to

$$\begin{aligned} \bar{E}(t) = & \frac{Z_0 dl}{4\pi} \left[\frac{1}{cR} \frac{\partial I}{\partial t} \left(t - \frac{R}{c} \right) + \frac{3}{R^2} I \left(t - \frac{R}{c} \right) + \frac{3c}{2R^3} \text{sgn}(t) \otimes I \left(t - \frac{R}{c} \right) \right] \frac{XZ}{R^2} \hat{x} , \\ & + \frac{Z_0 dl}{4\pi} \left[\frac{1}{cR} \frac{\partial I}{\partial t} \left(t - \frac{R}{c} \right) + \frac{3}{R^2} I \left(t - \frac{R}{c} \right) + \frac{3c}{2R^3} \text{sgn}(t) \otimes I \left(t - \frac{R}{c} \right) \right] \frac{YZ}{R^2} \hat{y} \\ & + \frac{Z_0 dl}{4\pi} \left[\frac{1}{cR} \frac{\partial I}{\partial t} \left(t - \frac{R}{c} \right) \frac{Z^2 - R^2}{R^2} + \left[\frac{1}{R^2} I \left(t - \frac{R}{c} \right) + \frac{c}{2R^3} \text{sgn}(t) \otimes I \left(t - \frac{R}{c} \right) \right] \frac{3Z^2 - R^2}{R^2} \right] \hat{z} , \end{aligned} \quad (12)$$

$$\bar{H}(t) = \frac{dl}{4\pi} \left[\frac{1}{cR} \frac{\partial I}{\partial t} \left(t - \frac{R}{c} \right) + \frac{1}{R^2} I \left(t - \frac{R}{c} \right) \right] \left(-\frac{Y}{R} \hat{x} + \frac{X}{R} \hat{y} \right) , \quad (13)$$

where $\text{sgn}(t) = \frac{t}{|t|}$, $t \neq 0$, and \otimes is a convolution operator.

We can change the orientation of the dipole so that it matches those of AESOP by rotating the coordinates as indicated in figure 6(c):

$$\begin{array}{ccc} \text{from } x & \text{to } y \\ y & \text{to } z \\ z & \text{to } x \end{array}$$

Finally, the electric and magnetic fields of a dipole are calculated by

$$\begin{aligned}\bar{E}(t) = & \frac{Z_0 dl}{4\pi} \left[\frac{1}{cR} \frac{\partial I}{\partial t} \left(t - \frac{R}{c} \right) \frac{X^2 - R^2}{R^2} + \left[\frac{1}{R^2} I \left(t - \frac{R}{c} \right) + \frac{c}{2R^3} \text{sgn}(t) \otimes I \left(t - \frac{R}{c} \right) \right] \frac{3X^2 - R^2}{R^2} \right] \hat{x} \\ & + \frac{Z_0 dl}{4\pi} \left[\frac{1}{cR} \frac{\partial I}{\partial t} \left(t - \frac{R}{c} \right) + \frac{3}{R^2} I \left(t - \frac{R}{c} \right) + \frac{3c}{2R^3} \text{sgn}(t) \otimes I \left(t - \frac{R}{c} \right) \right] \frac{XY}{R^2} \hat{y} \\ & + \frac{Z_0 dl}{4\pi} \left[\frac{1}{cR} \frac{\partial I}{\partial t} \left(t - \frac{R}{c} \right) + \frac{3}{R^2} I \left(t - \frac{R}{c} \right) + \frac{3c}{2R^3} \text{sgn}(t) \otimes I \left(t - \frac{R}{c} \right) \right] \frac{XZ}{R^2} \hat{z} ,\end{aligned}\quad (14)$$

$$\bar{H}(t) = \frac{dl}{4\pi} \left[\frac{1}{cR} \frac{\partial I}{\partial t} \left(t - \frac{R}{c} \right) + \frac{1}{R^2} I \left(t - \frac{R}{c} \right) \right] \left(-\frac{Z}{R} \hat{y} + \frac{Y}{R} \hat{z} \right). \quad (15)$$

4. Implementation

The dipole current distribution and field equations developed in section 3 provide us the means to compute the free field emitted by AESOP. But the total field in the half space above the ground is a result of the superposition of the free field and the reflected field from the ground.

To simplify the problem, we assume that the earth ground is a perfectly electrically conducting (PEC) half space. With this assumption, we can apply the method of images to calculate the reflected field. The calculation can be further simplified if we use the symmetry of dipole currents through the plane that is perpendicular to the axis of AESOP at the center. The biconic and cylindrical parts of the antenna are assumed to be one single transmission line, having both ends slanted toward the ground.

We model the simulator as a series of dipole elements representing the physical structure of the AESOP simulator. The number of dipole elements has been selected to resolve the early-time structure of the field and to produce a relatively smooth waveform. For every 0.01-m section of the antenna, we evaluate the current distribution developed in section 3.1 at the time and spatial location that is the driving source for the corresponding dipole element. We employ 14,978 dipole elements with a dipole separation of 0.01 m to represent half the antenna of AESOP.

We let the coordinates be as depicted in figure 7. Thus, the $z = 0$ plane is the ground plane and parallel to AESOP. The method of images is applied through the ground plane to compute the reflected fields. The $y = 0$ plane contains the antenna. The symmetry of current distribution is through the $x = 0$ plane, called the center plane. The height of the antenna is 20 m above the ground. Each end of AESOP is terminated at 6.5 m above the ground. Each slanted section begins and ends at 100.4 and 147.9 m from the center plane, respectively. To save computational time, we translate and rotate the coordinate system to determine the fields of the simulator slanted sec-

tions, as depicted in figure 8. Finally, we use superposition of all fields to obtain the total field at an observation point from the AESOP simulator.

We have used a FORTRAN program to compute all fields at an observation point. The program is listed in appendix A.

Figure 7. Illustration of dipole model for AESOP simulator.




Figure 8. Coordinates for slanted dipoles.

5. Results and Comparison with Empirical Data

We implemented a FORTRAN program to compute the fields at an observation point out to 2 μ s in time with a time step of 1 ns. The following are the execution times of the program on some systems:

System	Execution time (hour:minute)
IBM 3090 with MVS/XA 2.3	4:00
IBM RISC/6000-530 with AIX 3.2.4	3:00
IBM RISC/6000-550 with AIX 3.2.4	2:18
IBM RISC/6000-560 with AIX 3.2.4	1:50
SGI 4D/440 with IRIX 4.0.5	2:32
SGI 4D/35 with IRIX 4.0.5	11:11
DECstation 5000 with ULTRIX 4.3	14:49

Because of the symmetry, the computed field components E_x , H_y , and H_z are the only nonzero components of the field on the center plane ($x = 0$ plane). Off the center plane, the computed model indicates that the H_x component is zero (in retarded time) until the current arrives to the slanted section of the simulator. In the plane containing the antenna ($y = 0$ plane), the symmetry requires that E_x , E_z , and H_y be the only nonzero fields. The calculated fields show the effects of the change in direction of the current on the antenna structure, as slope discontinuities in the fields, at the expected time.

For comparison to the empirical data, we scale all the computed fields by a factor that scales the computed total electric field of the point (0,50,20) to 50 kV/m. In comparison, we notice that the idealized lossless ground plane assumption has caused the peak fields of the model to be generally overestimated (compared to the empirical data) by roughly a factor of two; this assumption also led to the inaccuracy of the vertical field prediction from the slanted end sections of the simulator. Assuming a perfectly conducting ground plane means that the model does not reflect the effects of the pseudo-Brewster angle on the vertical fields. However, the energy in the predicted field waveforms still provides an upper bound for the actual fields.

Appendix B contains field plots compared to empirical data for the following sample points in the space:

(0, -176, 1)
(0, -176, 4.6)
(-332, -176, 1)
(-332, -176, 4.6)
(0, 1546, 4.6)
(410, 502, 4.6)
(-444, 1341, 4.6)
(473, 0, 1)
(473, 0, 4.6)

6. Conclusion

Although the perfectly electrically conducting ground assumption has generally overestimated and distorted the calculated field, the predicted fields are still good for bounding problems, especially for points high above the ground. The calculated information can provide upper bound contours or isosurfaces at certain field magnitudes. The number of dipoles used in the model can be decreased, with a corresponding decrease in the model run time, but with an increase in "spiky" noise in the predicted waveforms. This model has been helpful in understanding the relationship between the characteristics of the field waveforms and the structure of the AESOP simulator. Because it is simple and requires only 2.5 MB of memory to execute, this model can be used on most computer platforms. This code is the first building block to develop more accurate field predictions for the AESOP simulator.

Appendix A.—Program Listing

```

      IMPLICIT NONE
      INTEGER*4 MAXF, MAXS, MAXT, MAXSH
      PARAMETER (MAXF=8, MAXS=14978, MAXSH=10040, MAXT=2000)
      CHARACTER DSCUCO*80, DSEHD*80, DSEHT*80, SPC*1
      REAL*8 COEF(20,MAXS), CUC
      REAL*8 CUCC, CUL, CUR
      REAL*8 EH(0:MAXT,MAXF), RL(4), RO(12), RS(4)
      REAL*8 EHM(8), ROS(4), ROT(4)
      REAL*8 OXT, OZT, SA, SAX, SAZ, SINA, COSA
      REAL*8 AL, C, DL, DLMIN, DT, PI, ZO, ZOO4PI
      REAL*8 CU, CUD, CVO, ETC, HTC, TCS, TEHS, TR, RETARD
      REAL*8 CUP, TCRS, SN, SN1, SNR, SNR1
      INTEGER*4 IB, INF, IT, ITS, IS, ISB, ISE, ISL, ISGX, ISGZ
      INTEGER*4 I, J, K, ITM(8), ISGXB, ISGXE, ISGZB, ISGZE
      COMMON /COEFBL/COEF,CUC
      COMMON /CUPARM/CUP,TCRS,SN,SN1,SNR,SNR1
      DATA CUCC,CUL,CUR/0.273448D-12,4.07D-9,5.364D-3/

*
      CUP = DSQRT(CUL*CUCC)
      CUC = CUR/(2.*DSQRT(CUL/CUCC))

*
* INITIALIZE PARAMETERS
*
      ISGXB = 1
      ISGXE = -1
      ISGZB = 1
      ISGZE = -1
      SPC = '^'
C      C = 2.997925D8
      C = 3.D8
      PI = 4.*DATAN(1.D0)
      ZO = 4.*PI*C*1.D-7
C      ZO = 376.991184D0
      ZOO4PI=C*1.D-7
      DLMIN = 1.D-2
      DT = 1.D-9
      AL = 20.
      RS(2) = 0.
      OXT = 100.40D0
      OZT = AL
      SAX = 47.5D0
      SAZ = 13.5D0
      SA = DSQRT(SAX**2 + SAZ**2)

```

```

        COSA = SAX/SA
        SINA = SAZ/SA
        ISE = 0
*
* READ IN CURRENT COEFFICIENTS OF ALL SEGMENTS
*
        DSCUCO = '/home/emsc3/bluu/aesop/CUCOE.F.DAT'
        OPEN(28, FILE=DSCUCO, STATUS='OLD', FORM='FORMATTED')
        DO 170 J=1, MAXS
            READ(28,*) IS, (COEF(I,J), I=1,20)
170 CONTINUE
        CLOSE(28)
*
* READ PARTITION DATA SET NAME OF THE E AND H FIELD
*
        READ(5, '(A)') DSEHD
        IB = INDEX(DSEHD, ' ')
        IB = IB - 1
*
* READ IN SEGMENT LENGTH AND COORDINATE OF OBSERVATION POINT
*
115 READ(5,*,END=999) DL, (RO(I), I=1,12)
*
        ISL = (DL + DLMIN/2)/DLMIN
*
* DETERMINE THE STARTING TIME
*
        ISB = (ISL+1)/2
        ISGX = 1
        IF (RO(1).LT.0.) ISGX = -1
        RS(1) = ISGX*(2*ISB-1)*(DLMIN/2.)
        TEHS = DSQRT((RO(1)-RS(1))**2 + RO(2)**2 + (RO(3)-AL)**2)/C +
        | ISB*CUP
*
* INITIALIZE EH ARRAY
*
        DO 150 J=1, MAXF
            DO 150 I=0, MAXT
                EH(I,J) = 0.
150 CONTINUE
*
* CALCULATE E AND H FIELD
*
        WRITE(6,*) 'P', RO(1), RO(2), RO(3)
        WRITE(6,*) ' DIPOLE LENGTH IS', DL

```

```

DO 200 IS=ISB, MAXSH, ISL
  ISE = IS
  TCS = IS*CUP
  TCRS = (2*(MAXS-IS)+1)*CUP
  SN = IS-1
  SN1 = IS
  SNR = 2*MAXS - IS
  SNR1= 2*MAXS - IS +1
DO 220 ISGZ= ISGZB,ISGZE,-2
DO 220 ISGX= ISGXB,ISGXE,-2
  RS(1) = ISGX*(2*IS-1)*(DLMIN/2.)
  RS(3) = ISGZ*AL
C  RS(4) = DSQRT(RS(1)**2 + RS(2)**2 + RS(3)**2)
  RL(1) = RO(1) - RS(1)
  RL(2) = RO(2) - RS(2)
  RL(3) = RO(3) - RS(3)
  RL(4) = DSQRT(RL(1)**2 + RL(2)**2 + RL(3)**2)
*
  RETARD = RL(4)/C + TCS - TEHS
  ITS = RETARD/DT + 1.
DO 250 IT=ITS, MAXT
  TR = IT*DT - RETARD
  CALL CURRENT(IS,TR,CU,CUD,CVO)
  HTC = ISGZ*DL*(CUD/RL(4)/C +
    | CU/RL(4)**2)/(4.*PI)
*  EH(IT,4) = EH(IT,4) + 0.
  EH(IT,5) = EH(IT,5) - RL(3)/RL(4)*HTC
  EH(IT,6) = EH(IT,6) + RL(2)/RL(4)*HTC
  ETC = ISGZ*ZOO4PI*DL*(CUD/RL(4)/C +
    | 3.*(CU/RL(4)**2) + (1.5*C)*(CVO/RL(4)**3))
  EH(IT,1) = EH(IT,1) + ISGZ*ZOO4PI*DL*(
    | CUD/RL(4)/C*((RL(1)/RL(4))**2-1.) +
    | (CU/RL(4)**2+(C/2.)*(CVO/RL(4)**3))*
    | (3.*(RL(1)/RL(4))**2-1.) )
  EH(IT,2) = EH(IT,2) + RL(1)/RL(4)*RL(2)/RL(4)*ETC
  EH(IT,3) = EH(IT,3) + RL(1)/RL(4)*RL(3)/RL(4)*ETC
250  CONTINUE
220  CONTINUE
200  CONTINUE
*
*  CALCULATE THE FIELD CONTRIBUTE BY THE SLANT PARTS
*
  RS(3) = 0.
  ROT(2) = RO(2)
  ISB = ISE + ISL

```

```

DO 260 IS=ISB, MAXS, ISL
  TCS = IS*CUP
  TCRS = (2*(MAXS-IS)+1)*CUP
  SN = IS-1
  SN1 = IS
  SNR = 2*MAXS - IS
  SNR1= 2*MAXS - IS +1
DO 280 ISGZ= ISGZB,ISGZE,-2
DO 280 ISGX= ISGXB,ISGXE,-2
  ROS(1) = RO(1) - ISGX*OXT
  ROS(3) = RO(3) - ISGZ*OZT
  ROT(1) = COSA*ROS(1) - ISGX*ISGZ*SINA*ROS(3)
  ROT(3) = ISGX*ISGZ*SINA*ROS(1) + COSA*ROS(3)
  RS(1) = ISGX*(2*(IS-MAXSH)-1)*(DLMIN/2.)
C  RS(4) = DSQRT(RS(1)**2 + RS(2)**2 + RS(3)**2)
  RL(1) = ROT(1) - RS(1)
  RL(2) = ROT(2) - RS(2)
  RL(3) = ROT(3) - RS(3)
  RL(4) = DSQRT(RL(1)**2 + RL(2)**2 + RL(3)**2)
*
  RETARD = RL(4)/C + TCS - TEHS
  ITS = RETARD/DT + 1.
DO 290 IT=ITS, MAXT
  TR = IT*DT - RETARD
  CALL CURRENT(IS,TR,CU,CUD,CVO)
*
  HTC = ISGZ*DL*(CUD/RL(4)/C +
    CU/RL(4)**2)/(4.*PI)
  EHM(6) = RL(2)/RL(4)*HTC
*
  EH(IT,4) = EH(IT,4) + ISGX*ISGZ*SINA*EHM(6)
  EH(IT,5) = EH(IT,5) - RL(3)/RL(4)*HTC
  EH(IT,6) = EH(IT,6) + COSA*EHM(6)
*
  ETC = ISGZ*ZOO4PI*DL*(CUD/RL(4)/C +
    3.*(CU/RL(4)**2) + (1.5*C)*(CVO/RL(4)**3))
  EHM(1) = ISGZ*ZOO4PI*DL*(
    CUD/RL(4)/C*((RL(1)/RL(4))**2-1.) +
    (CU/RL(4)**2+(C/2.)*(CVO/RL(4)**3))*
    (3.*(RL(1)/RL(4))**2-1.)
  )
  EHM(3) = RL(1)/RL(4)*RL(3)/RL(4)*ETC
*
  EH(IT,1) = EH(IT,1) + COSA*EHM(1)+ISGX*ISGZ*SINA*EHM(3)
  EH(IT,2) = EH(IT,2) + RL(1)/RL(4)*RL(2)/RL(4)*ETC
  EH(IT,3) = EH(IT,3) - ISGX*ISGZ*SINA*EHM(1)+COSA*EHM(3)

```

```

290      CONTINUE
280      CONTINUE
260 CONTINUE
*
*
      DO 270 I=1,8
          EHM(I) = 0.
          ITM(I) = 0.
270 CONTINUE
*
* DETERMINE THE MAX VALUES OF E & H FIELDS AND THEIR TIMES
*
      DO 295 IT=0, MAXT
          ETC = DSQRT(EH(IT,1)**2 + EH(IT,2)**2 + EH(IT,3)**2)
          HTC = DSQRT(EH(IT,4)**2 + EH(IT,5)**2 + EH(IT,6)**2)
          EH(IT,7)=ETC
          EH(IT,8)=HTC
          IF (ETC.GT.EHM(7)) THEN
              EHM(7) = ETC
              ITM(7) = IT
          ENDIF
          IF (HTC.GT.EHM(8)) THEN
              EHM(8) = HTC
              ITM(8) = IT
          ENDIF
          DO 297 I=1,6
              IF (ABS(EH(IT,I)).GT.ABS(EHM(I))) THEN
                  EHM(I) = EH(IT,I)
                  ITM(I) = IT
              ENDIF
297      CONTINUE
295 CONTINUE
*
* SET A DATA SET NAME FOR E AND H FIELD
*
      WRITE(DSEHT, '(SP,A,E14.7,E14.7,E14.7)') DSEHD(1:IB),
| RO(1), RO(2), RO(3)
      IF(DSEHT(IB+1:IB+1).NE.' ') THEN
          DSEHT(IB+2:IB+2) = DSEHT(IB+1:IB+1)
          DSEHT(IB+16:IB+16) = DSEHT(IB+15:IB+15)
          DSEHT(IB+30:IB+30) = DSEHT(IB+29:IB+29)
      ENDIF
      DSEHT(IB+1:IB+1) = SPC
      DSEHT(IB+15:IB+15) = SPC
      DSEHT(IB+29:IB+29) = SPC

```

```

      RETARD = TEHS/DT
*
* WRITE E AND H FIELD TO FILE
*
      OPEN(32, FILE=DSEHT, STATUS='UNKNOWN', FORM='UNFORMATTED')
      WRITE(32) RO(1),RO(2),RO(3),RO(4),RO(5),RO(6),DL,DT
      WRITE(32) RO(7),RO(8),RO(9),RO(10),RO(11),RO(12),TEHS,RETARD
      WRITE(32) EHM(1),EHM(2),EHM(3),EHM(4),EHM(5),EHM(6),EHM(7),
| EHM(8)
      WRITE(32) ITM(1),ITM(2),ITM(3),ITM(4),ITM(5),ITM(6),
| ITM(7),ITM(8)
      DO 400 I=0, MAXT
          WRITE(32) (EH(I,J), J=1, 8)
400 CONTINUE
      CLOSE(32)
*
      WRITE(6,*) 'END OF THIS RUN; DATA IN FILE: ', DSEHT
      GOTO 115
999 STOP
      79 FORMAT(8E15.7)
      END
*
*
** SUBROUTINE CALCULATE THE DIPOLE'S
** CURRENT,
** CURRENT DERIVATIVE,
** CONVOLUTION OF SIGN FUNCTION AND CURRENT
*
*
      SUBROUTINE CURRENT(NS,TS,Y,YD,YCO)
      IMPLICIT NONE
      INTEGER*4 MAXS, NS, LR
      REAL*8 P, TC2, SN, SN1, SNR, SNR1
      REAL*8 T, TS, Y, YS, YD, YSD, YCO, YSCO
      COMMON /CUPARM/P,TC2,SN,SN1,SNR,SNR1
C      DATA P/0.333606558688524581D-10/
C      DATA C/0.219836066437990879D-04/
*
*
      LR = 0
      T = TS
      CALL CURDIS(NS,LR,SN1,T,YS,YSD,YSCO)
      T = TS + P
      CALL CURDIS(NS,LR,SN,T,Y,YD,YCO)
      Y = Y - YS

```

```

      YD = YD - YSD
      YCO = YCO - YSCO
*
* REFLECTION CURRENT IS NOT EMERGING
*
      IF (T.LE.TC2) RETURN
      T = TS - TC2
*
* REFLECTION CURRENT
*
      LR = 12
      CALL CURDIS(NS,LR,SNR1,T,YS,YSD,YSCO)
      Y = Y + YS
      YD = YD + YSD
      YCO = YCO + YSCO
      T = T + P
      CALL CURDIS(NS,LR,SNR,T,YS,YSD,YSCO)
      Y = Y - YS
      YD = YD - YSD
      YCO = YCO - YSCO
      RETURN
      END
*
*
** SUBROUTINE CALCULATES CURRENT PARAMETERS **
*
*
      SUBROUTINE CURDIS(NS,IR,SN,T,YS,YSD,YSCO)
      IMPLICIT NONE
      INTEGER*4 MAXS, NS, IR
      PARAMETER (MAXS=14978)
      REAL*8 CO(20,MAXS), C
      REAL*8 T, SN, YS, YSD, YSCO
      COMMON /COEFBL/CO,C
*
* AA = CO( 1+IR,NS)
* AB = CO( 2+IR,NS)
* AC = CO( 3+IR,NS)
* AD = CO( 4+IR,NS)
* PD = CO( 5+IR,NS)
* AF = CO( 6+IR,NS)
* EA = CO( 7,NS)
* EB = CO( 8,NS)
* EC = CO( 9,NS)
* ED = CO(10,NS)

```



```

*      EDP = CO(11,NS)
*      EF  = CO(12,NS)
*      RA  = CO(13,NS)
*      RB  = CO(14,NS)
*      RC  = CO(15,NS)
*      RD  = CO(16,NS)
*      PDR = CO(17,NS)
*      RF  = CO(18,NS)
*      AR  = CO(19,NS)
*      ER  = CO(20,NS)
*
*
* CURRENT DISTRIBUTION
*
      YS = DEXP(-SN*C)*( CO(1+IR,NS)*DEXP(CO(7,NS)*T) +
      |                  CO(2+IR,NS)*DEXP(CO(8,NS)*T) +
      |                  CO(3+IR,NS)*DEXP(CO(9,NS)*T) +
      |                  CO(6+IR,NS)*DEXP(CO(12,NS)*T) +
      | 2.*CO(4+IR,NS)*DEXP(CO(10,NS)*T)*DCOS(CO(11,NS)*T+CO(5+IR,NS)) )
*
* DERIVATIVE OF CURRENT DISTRIBUTION
*
      YSD = DEXP(-SN*C)*( CO(1+IR,NS)*CO(7,NS)*DEXP(CO(7,NS)*T) +
      |                  CO(2+IR,NS)*CO(8,NS)*DEXP(CO(8,NS)*T) +
      |                  CO(3+IR,NS)*CO(9,NS)*DEXP(CO(9,NS)*T) +
      |                  CO(6+IR,NS)*CO(12,NS)*DEXP(CO(12,NS)*T) +
      | 2.*CO(4+IR,NS)*DEXP(CO(10,NS)*T)*(
      |                  CO(10,NS)*DCOS(CO(11,NS)*T+CO(5+IR,NS)) -
      |                  CO(11,NS)*DSIN(CO(11,NS)*T+CO(5+IR,NS)) ) )
*
* CONVOLUTION OF SGN(T) AND I(T)
*
      YSCO = DEXP(-SN*C)*(
      | CO(1+IR,NS)*(2.*DEXP(CO(7,NS)*T)-1.)/CO(7,NS) +
      | CO(2+IR,NS)*(2.*DEXP(CO(8,NS)*T)-1.)/CO(8,NS) +
      | CO(3+IR,NS)*(2.*DEXP(CO(9,NS)*T)-1.)/CO(9,NS) +
      | CO(6+IR,NS)*(2.*DEXP(CO(12,NS)*T)-1.)/CO(12,NS) +
      | 2.*CO(4+IR,NS)*( 2.*DEXP(CO(10,NS)*T)*
      |                  (CO(10,NS)*DCOS(CO(11,NS)*T+CO(5+IR,NS)) +
      |                  CO(11,NS)*DSIN(CO(11,NS)*T+CO(5+IR,NS)) ) -
      |                  (CO(10,NS)*DCOS(CO(5+IR,NS)) +
      |                  CO(11,NS)*DSIN(CO(5+IR,NS)) ) ) /
      | (CO(10,NS)**2+CO(11,NS)**2) )
      IF(IR.EQ.0) RETURN
      YS= YS + DEXP(-SN*C)*CO(19,NS)*DEXP(CO(20,NS)*T)

```

```
YSD= YSD + DEXP(-SN*C)*CO(19,NS)*CO(20,NS)*DEXP(CO(20,NS)*T)
YSCO= YSCO + DEXP(-SN*C)*CO(19,NS)*(2.*DEXP(CO(20,NS)*T)-1.)/
|      CO(20,NS)
RETURN
END
```

Appendix B.—Field Plot Comparison

The field plots presented display electric or magnetic field component waveforms at selected test points, showing both empirical measurements and computed results in the same graph. No comparisons are done for field components that are zero in theory. Both of the fields, empirical and computed, are scaled to the same reference field level, 50 kV/m at 50 m away from the center of AESOP on the center plane. We selected the field points based on the availability of empirical data for test points that were measured for the Perimeter Field Map at the Woodbridge Research Facility. For those test points that have more than one measurement, we have chosen the best data set to compare with computed data. The selected field comparison points are as follows:

(0, -176, 1)	(fig. B-1)
(0, -176, 4.6)	(fig. B-2)
(-332, -176, 1)	(fig. B-3)
(-322, -176, 4.6)	(fig. B-4)
(0, 1546, 4.6)	(fig. B-5)
(410, 502, 4.6)	(fig. B-6)
(-444, 1341, 4.6)	(fig. B-7)
(473, 0, 1)	(fig. B-8)
(473, 0, 4.6)	(fig. B-9)

Figure B-1. Field comparison at (0,-176,1) meters: (a) E_x , (b) H_y .





Figure B-1 (cont'd).
Field comparison at
(0,-176,1) meters:
(c) H_z .




Figure B-2. Field comparison at (0,-176,4.6) meters:
(a) E_x , (b) H_y .





Figure B-2 (cont'd).
Field comparison at
(0,-176,4.6) meters:
(c) H_z .

(c)




Figure B-3. Field comparison at (-332,-176,1) meters:
(a) E_x , (b) E_y .





Figure B-3 (cont'd).
Field comparison at
(-332,-176,1) meters:
(c) E_z , (d) H_x .





Figure B-3 (cont'd).
Field comparison at
(-332,-176,1) meters:
(e) H_y , (f) H_z .





Figure B-4. Field comparison at (-332,-176,4.6) meters:
(a) E_x , (b) E_y .





Figure B-4 (cont'd).
Field comparison at
(-332,-176,4.6) meters:
(c) E_z , (d) H_x .





Figure B-4 (cont'd).
Field comparison at
(-332,-176,4.6) meters:
(e) H_y , (f) H_z .





Figure B-5. Field comparison at (0,1546,4.6) meters:
(a) E_x , (b) H_y .




Figure B-5 (cont'd).
Field comparison at
(0,1546,4.6) meters:
(c) H_z .




Figure B-6. Field comparison at (410,502,4.6) meters:
(a) E_x , (b) E_y .





Figure B-6 (cont'd).
Field comparison at
(410,502,4.6) meters:
(c) E_z , (d) H_x .






Figure B-6 (cont'd).
Field comparison at
(410,502,4.6) meters:
(e) H_y , (f) H_z .

(e)



(f)




Figure B-7. Field comparison at (-444,1341,4.6) meters:
 (a) E_x (b) E_y .





Figure B-7 (cont'd).
Field comparison at
(-444,1341,4.6) meters:
(c) E_z , (d) H_x .





Figure B-7 (cont'd).
Field comparison at
(-444,1341,4.6) meters:
(e) H_y , (f) H_z .





Figure B-8. Field comparison at (473,0,1) meters: (a) E_x , (b) E_z .





Figure B-8 (cont'd). (c)
Field comparison at
(473,0,1) meters: (c) H_y .




Figure B-9. Field comparison at (473,0,4.6) meters:
(a) E_x , (b) E_z .






Figure B-9 (cont'd).
Field comparison at
(473,0,4.6) meters:
(c) H_y .



Distribution

Admnstr
Defns Techl Info Ctr
Attn DTIC-DDA (2 copies)
Cameron Sta Bldg 5
Alexandria VA 22304-6145

Director
Defns Intllgnc Agcy
Attn RTS-2A Techl Lib
Washington DC 20301

Defns Nuc Agcy
Attn RAEE Elect Effects Div
6801 Telegraph Rd
Alexandria VA 22310-3398

Cmdr
US Army ARDEC
Attn AMSTA-AR-AEC-IE N Svendsen
Picatinny Arsenal NJ 07806-5000

Cmdr
US Army ARDEC
Attn AMSTA-AR-CCL-D W Williams
Bldg 65 N
Picatinny Arsenal NJ 07806-5000

US Army AVRDEC
Attn AMSAT-R-EFM P Haselbauer
4300 Goodfellow Blvd
ST Louis MO 63120-1798

US Army BRDEC
Attn SATB-FGE J Ferric
Attn SATB-FGE T Childers
FT Belvoir VA 22060-5606

Commander
US Army Matl Cmnd
Attn AMCAM-CN
5001 Eisenhower Ave
Alexandria VA 22333-0001

Director
US Army Mis Cmnd (USAMICOM)
Attn AMSMI-RD-CS-R Documents
Redstone Arsenal AL 35898-5400

Cmdr
US Army MRDEC
Attn AMSMI-RD-ST-CM J Vandier
Huntsville AL 35898-5240

US Army Natick RDEC
Attn SATNC-SUSD-SHD A Murphy
Kansas Stret
Natick MA 01760-5018

Commander
US Army Nuc & Chem Agcy
Attn MONA-NU R Pfeffer
Attn MONA-TS Lib
7150 Heller Loop Rd Ste 101
Springfield VA 22150

Cmdr
US Army TARDEC
Attn AMSTA-ZT G Baker
Warren MI 48397-5000

Commander
US Army TECOM
Attn STERT-TE-E J Knaur
Redstone Technical Test Center
Huntsville AL 35898-8052

US Army TECOM Techl Dir Ofc
Attn AMSTE-TC-D R Bell
Aberdeen Proving Ground MD 21005

Commander
US Army White Sands Missile Range
Attn STEWS-NE J Meason
White Sands Missile Range NM 88002-5180

Nav Rsrch Lab
Attn Code 4820 Techl Info Div
4555 Overlook Ave SW
Washington DC 20375-5000

Commander
Nav Surfc Weapons Ctr
Attn Code E231 Techl Lib
Dahlgren VA 22448-5020

Natl Inst of Stand & Techlgy
Attn V Ulbrecht Rsrch Info Ctr
Rm E01 Bldg 101
Gaithersburg MD 20899

DoD Joint Spectrum Ctr
Attn CA J Word
120 Worthing Basin
Annapolis MD 21401

Distribution

US Army Rsrch Lab
Attn AMSRL-SL-CS J Beilfuss
Edgewood MD 21010-5423

US Army Rsrch Lab
Attn AMSRL-SL-CM D Farenwald
Aberdeen Proving Ground MD 21005-5068

US Army Rsrch Lab
Attn AMSRL-OP-SD-TA Mail & Records
Mgmt
Attn AMSRL-OP-SD-TL Tech Library
(3 copies)
Attn AMSRL-OP-SD-TP Tech Pub

US Army Rsrch Lab
Attn AMSRL-WT-N J Ingram
Attn AMSRL-WT-ND B Luu (5 copies)
Attn AMSRL-WT-ND C Le
Attn AMSRL-WT-ND J R Miletta
Attn AMSRL-WT-ND R J Chase
Attn AMSRL-WT-NF B Benwell
Attn AMSRL-WT-NF M H Mar
Attn AMSRL-WT-NF T Bock
Adelphi MD 20783-1197